Sample solutions of selected problems with Macaulay2 (Sets 2, 3)

Chris Peterson and Hirotachi Abo

Problem 4 (Set 2). Let `R` be the polynomial ring with variables `x` and `y` over rational numbers `QQ`.

```
i1 : R=QQ[x,y]

o1 = R

o1 : PolynomialRing
```

Define the ideal generated by x^2 and xy:

```
i2 : I=ideal(x^2,x*y)

              2
o2 = ideal (x , x*y)

o2 : Ideal of R
```

Compute the radical of `I`:

```
i3 : radI=radical I

o3 = ideal x

o3 : Ideal of R
```

In `Macaulay2`, you can use the function `primaryDecomposition` to find the primary decomposition of an ideal:

```
i4 : compI=primaryDecomposition(I)

                                                 2
o4 = {monomialIdeal x, monomialIdeal (x , y)}

o4 : List
```

This command returns the list of primary ideals. Their intersection is equal to `I`. Let `I1` be the first ideal in `compI` and and let `I2` be the second ideal in `compI`.

```
i5 : I1=compI#0

o5 = monomialIdeal x

o5 : MonomialIdeal of R

i6 : I2=compI#1

                          2
o6 = monomialIdeal (x , y)

o6 : MonomialIdeal of R
```

Compute the intersect, denoted by `J`, of `I1` and `I2` with `intersect`:

```
i7 : J=intersect(I1,I2)

                          2
o7 = monomialIdeal (x , x*y)

o7 : MonomialIdeal of R
```

This output tells you that `I==J`.

Problem 5 (Set 2). Redefine the ideal:

```
i8 : I=ideal (x^2-1,y^2-4)

                  2       2
o8 = ideal (x  - 1, y  - 4)

o8 : Ideal of R
```

As in Problem 4, the primary decomposition can be computed by primary-Decomposition:

```
i9 : compI=primaryDecomposition(I)
     -- used 0.03 seconds
     -- used 0.01 seconds
     -- used 0.06 seconds

o9 = {ideal (y - 2, x + 1), ideal (y - 2, x - 1), ideal (y + 2, x + 1), ideal (

o9 : List
```

The output does not fit. So let's count the number of components. This is given as follows:

```
i10 : #compI

o10 = 4
```

You can see each component of `I` as in the previous problem. Clearly, they are maximal ideals.

Problem 6,7 (Set 2). Consider the following ideal:

```
i11 : I=ideal(x^2,x*y^2,y^3)

               2     2   3
o11 = ideal (x , x*y , y )

o11 : Ideal of R
```

Define the quotient ring `R/I` as follows:

```
i12 : Q=R/I

o12 = Q

o12 : QuotientRing
```

Note that $V(\mathtt{I})$ is zero-dimensional, because its radical is (x, y). So Q is a finite-dimensional vector space. A basis for this vector space can be computed with basis:

```
i13 : basis Q

o13 = | 1 x xy y y2 |

               1       5
o13 : Matrix Q  <--- Q
```

Replaciing `I` by $(y^2 - x^2, y^2 + x^2)$ and taking the same steps give you the answer to Problem 7.

Problem 12 (Set 3). The set V of the three points is defined in affine 3-space. Define the corresponding polynomial ring:

```
i14 : S=QQ[x,y,z]

o14 = S

o14 : PolynomialRing
```

Define the corresponding ideals:

```
i15 : I1=ideal(x-1,y-2,z-3)

o15 = ideal (x - 1, y - 2, z - 3)

o15 : Ideal of S

i16 : I2=ideal(x-1,y-3,z-7)

o16 = ideal (x - 1, y - 3, z - 7)

o16 : Ideal of S

i17 : I3=ideal(x-2,y-3,z-5)

o17 = ideal (x - 2, y - 3, z - 5)

o17 : Ideal of S
```

Compute the intersection of these ideals:

```
i18 : I=intersect(I1,I2,I3)

               1  3   15  2   71     105    1     2   3        1  2   35          3
o18 = ideal (-*z  - --*z  + --*z - ---, - -*y*z  + -*y*z + -*z  - --*y - 3z + -
               8       8       8      8     8         2        4      8

o18 : Ideal of S
```

Is this ideal radical?

```
i19 : I==radical I

o19 = true
```

So `I` is equal to $I(V)$.